

The World is Not Enough: Base SAS® Visualizations and Geolocations

Louise S. Hadden, Cormac Corporation

ABSTRACT

Geographic processing in SAS® has recently undergone some major changes: as of Version 9.4 Maintenance Release M5 many procedures formerly a part of SAS/Graph are now available in BASE SAS. At the same time, SAS Graphics have added some new procedures such as PROC SGMAP that build on the functionality of SAS/GRAPH's PROC GMAP and incorporate ODS graphics techniques including attribute maps and image annotation. This paper and poster will replicate a map of the world created by the author with SAS/GRAPH and PROC GMAP with the annotate facility using PROC SGMAP to map three different metrics on a map of the world. New SAS mapping and SG procedure techniques will be demonstrated, following Agent 007's adventures across the globe.

INTRODUCTION

The focus of this paper is procedures, tools and techniques that create geographic visualizations in SAS, emphasizing opportunities for customization. Mapping in SAS was once accomplished with SAS/Graph (and prior to that, SAS/GIS); more recently, ODS graphics, in particular the SGPLOT and SGMAP procedures and templates available in BASE SAS are being utilized. Enhancing output from SAS/GRAPH has been the subject of many SAS papers over the years, including my own, and many excellent techniques have been explored and developed, and will continue to be. Maps created from ODS graphics are often "camera-ready" without any user intervention, but on occasion there is a need for additional customization, and ODS graphics has many tools available for just that purpose. This paper and e-poster undertakes to demonstrate the similarities and differences by demonstrating the recreation of a world map portraying population density that was pioneered decades ago with the newer ODS Graphics techniques, and then retools the code to follow 007's exploits across the globe. The original image created was called "World at Night", modified from a sample provided by Robert Allison. (Hadden, 2018). Techniques for enhancing visualizations such as ATTR (attribute) assignments including ATTRMAPS and annotation will be demonstrated in the process. This paper and e-poster are suitable for all levels of expertise.

BASIC REQUIREMENTS FOR MAPPING

The essential ingredients for a SAS-produced map include a map data set with geographic IDs, latitude, and longitude; and a response data set which includes geographic IDs and the information to be portrayed on a map. Annotation data sets, attribute settings, and graphic option sets can enhance maps created with SAS in both SAS/GRAPH and ODS GRAPHICS.

MAP DATA SET

SAS/Graph currently provides access to GFK marketing maps as part of the SAS/Graph license. These maps are high-quality, commercially licensed geographic boundary maps which are significantly more detailed and current than the free, sometimes user-produced, data sets that came with older SAS releases. (Hadden, 2017). GFK maps are digital boundary files that represent administrative, postal, and statistical regions across the world. They have more up-to-date boundaries and cover more countries than the original. Moreover, they have a consistent coding convention, and each boundary file comes with a matching attribute file containing shape names, etc. ODS Graphics users may or may not have a SAS/Graph license; there are many "free" shape files of different geopolitical areas available on the internet, and a BASE SAS procedure to import ESRI shape files, PROC MAPIMPORT. (Hadden, 2016.) GFK map files are also typically "pre-projected" with latitude and longitude. ODS Graphics users without access to GFK map files may need to project their map data sets, and those who are using annotate data bases will need to project their map data sets along with their response data to ensure a match. This can be done using the BASE SAS procedure PROC GPROJECT.

RESPONSE DATA SET

A response data set is a SAS data set with geographic IDs that match the map file, with information to be visualized on the map. This data set can be completely external, or, in the case of GFK maps, can be the ATTR map, or, in the case of maps created by using PROC MAPIMPORT (ESRI SHP files), in the accompanying DBF. Files should have x and y coordinates, and/or latitude and longitude; or should match via the geographic ID (for example, a state code for a statistic collected at the state level.)

ANNOTATE DATA SETS

An annotate data set contains instructions to add custom graphics elements, such as text labels, shapes, lines, or symbols, to maps or plots, allowing for detailed, user-defined annotations. An annotate data set is linked to graphical output via ANNOTATE= in SAS/GRAPH, and via SGANNOTATE= in ODS Graphics.

A typical SAS/Graph annotate data set comprises a list of drawing commands, with each row specifying what to draw, where, and how. Core variables are function (e.g., label, move, draw, etc.), X & Y coordinates, text, color, size, style, position, and HSYS, XSYS, and YSYS. You can add labels, draw custom lines or arrows, highlight circles or shapes, add legends or callouts, and overlay symbols or logos. Annotate data sets require precise control over coordinates and drawing order and only work with SAS/GRAPH procedures.

SGAnnotate uses a data driven approach to add graphical elements directly to SG procedure output. Like SAS/Graph annotate, you create a SAS data set with annotation instructions. It works with SG procedures including SGMAP. Annotations align with plot axes unless using wall-percentage coordinates. Like SAS/GRAPH annotate, it supports layering. It uses ODS styles. The basis structure of an SGAnnotate data set includes an ID, DATATRANSARENCY, coordinates for lines and/or text, coordinate system (DATAVALUE, etc.), text, width, height, colors, line attributes, and text font and styling. Some benefits of SGAnnotate is that it is style-based and aligns with ODS, and allows a rich variety of style control including transparency, fonts, etc.

GRAPHIC OPTIONS AND ATTRIBUTE SETTINGS

As with annotate data sets, both SAS/GRAPH and ODS Graphics use graphic options and attributes. SAS/GRAPH uses GOPTIONS, and in the context of maps, you can use pattern assignments to specify fill color, and other in line options to assign line width and color, etc. ODS Graphics uses an ODS graphics statement, in which you can set global graph size and appearance, set a graph style, and work with templates. Within the SG procedure, one can control the control line, marker, and fill attributes (via STYLEATTRS) and set axis or title font sizes and styles (LABELATTRS, VALUEATTRS, TEXTATTRS). Conveniently, the ATTRS statements are consistent within the ODS graphics system.

MAPPING THE WORLD WITH SAS/GRAPH PROC GMAP

Below follows code snippets for creating the SAS/GRAPH version of a stylized World Population map. This map relies on an annotate data set with points for each city. This creates an effect resembling an aerial view of the world at night.

```
/* Create the annotate data base */
/* Obtain latitude and longitude from the MAPSGFK.WORLD_CITIES data set */
/* creates the dot_anno data set, which provides a dot for the centroid of each city */
/* dither the dots for some areas of Asia to make less dense */

data dot_anno;
    set mapsgfk.world_cities;
    anno_flag=1;

    if idname='China' then
        do;
            foo=ranuni(123);

            if foo>.95 then
                output;
        end;
```

```

        else if idname='Taiwan, Province of China' then
            do;
                foo=ranuni(123);

                if foo>.99 then
                    output;
            end;
        else if idname='South Korea' then
            do;
                foo=ranuni(123);

                if foo>.97 then
                    output;
            end;
        else if idname='Philippines' then
            do;
                foo=ranuni(123);

                if foo>.97 then
                    output;
            end;
        else
            do;
                output;
            end;
run;

/* Get the world map dataset mymap, and remove Antartica from the display */
set mapsgfk.world (where=((density<=1) and (idname^='Antarctica')))
drop = resolution);
run;

/* PROJECT THE MAP AND ANNOTATE DATA BASES TOGETHER */

/* project the map and the dots */
data combined;
    set mymap dot_anno;
run;

proc gproject data=combined out=combined latlong eastlong degrees dupok project=robinson;
    id id;
run;

/* SPLIT OUT THE TWO DATA BASES AFTER PROJECTION */

data mymap dot_anno;
    set combined;

    if anno_flag=1 then
        output dot_anno;
    else output mymap;
run;

/* Add additional details to the projected annotate data base */
/* Do this after the gproject, or the character variables will slow down gproject */

data dot_anno;
    set dot_anno;
    xsys='2';
    ysys='2';
    hsys='3';
    when='a';
    function='point';
    color='cornsilk';
    position='5';
run;

```

```

/* Set GOPTIONS */

goptions device=png;
goptions xpixels=1700 ypixels=800;
goptions cback=darkblue;

/* Open HTML file */

ODS LISTING CLOSE;
ODS HTML path=odsout body="&name..html"
      (title="SAS/Graph night sky simulation")
      nogtitle style=minimal;
title1 f="albany amt/bold" h=14pt "2024 World Population Density";
title2 f="albany amt" h=12pt "Using SAS' GFK maps and world cities data";

/* Set fill pattern and color */
pattern1 v=msolid c=black;

/* add labels for hmtl hover */

data mymap_data;
  set mapsgfk.world_attr;
  length my_html $300;
  my_html='title='||quote(trim(left(idname)));
  colorvar=1;
run;

/* create map */

proc gmap map=mymap data=mymap_data anno=dot_anno;
  id id;
  choro colorvar / levels=1 nolegend
    coutline=gray
    html=my_html
    des='' name="&name";
run;

quit;

ODS HTML CLOSE;
ODS LISTING;

```

Program 1. PROC GMAP version of World Population Map




Figure 1. World Population via PROC GMAP

MAPPING THE WORLD WITH BASE SAS PROC SGMAP

The PROC SGMAP version of the program creates a remarkably similar map, but with quite different syntax. The code needs to be run with BASE SAS 9.4 M7 or higher.

```
/* Get the world map */
data my_map;
  set mapsgfk.world
    (where=((density<=1) and (idname^='Antarctica')) drop = resolution);
run;

/* project the map */
proc gproject data=my_map out=my_map latlong eastlong degrees dupok project=robinson
  parmout=projparm;
  id id;
run;

/* drop the unprojected lat/long, so PROC SGMAP will not use them by default */
data my_map;
  set my_map (drop=lat long);
run;

/* Create the sgannotate data base */
/* Obtain latitude and longitude from the MAPSGFK.WORLD_CITIES data set */
/* creates the dot_anno data set, which provides a dot for the centroid of each city */
/* dither the dots for some areas of Asia to make less dense */

data white_dots;
  set mapsgfk.world_cities;

  if idname='China' then
    do;
      foo=ranuni(123);

      if foo>.95 then
        output;
    end;
  else if idname='Taiwan, Province of China' then
    do;
      foo=ranuni(123);

      if foo>.99 then
        output;
    end;
  else if idname='South Korea' then
    do;
      foo=ranuni(123);

      if foo>.97 then
        output;
    end;
  else if idname='Philippines' then
    do;
      foo=ranuni(123);

      if foo>.97 then
        output;
    end;
  else
    do;
      output;
    end;
run;
```

```

/* project the sgannotate */
proc gproject data=white_dots out=white_dots latlong eastlong degrees dupok
    parmin=projparm parmentry=my_map;
    id;
run;

data my_mapdata;
    set mapsgfk.world_attr;
    colorvar='1';
run;

/* Open HTML file */

ODS LISTING CLOSE;
ODS HTML path=odsout body="&name..htm"
    (title="World Population")
    style=htmlblue;

/* Set ODS Graphics */
ods graphics /
    noscale /* if you do not use this option, the text will be resized */
imagemap TIPMAX=31066
imagefmt=png imagename="&name"
antialias=off subpixel=off
width=1700px height=800px noborder;
title1 c=white h=16pt "World Population - PROC SGMAP ";

/* Create map */


proc sgmap maprespdata=my_mapdata mapdata=my_map noautolegend plotdata=white_dots;
    styleattrs datacolors=(black) backcolor=darkblue;
    choromap colorvar / mapid=idname lineattrs=(thickness=1 color=gray)
    tip=(idname) tiplabel=('Country');
    scatter x=x y=y / transparency=0 markerattrs=(size=1px color=cornsilk
symbol=squarefilled)
        tip=none;
run;

quit;

ODS HTML CLOSE;
ODS LISTING;

```

Program 2. PROC SGMAP version of World Population Map



THE WORLD OF JAMES BOND, AGENT 007

One of my guilty pleasures are James Bond books and movies. I used to take my babysitting money and buy secondhand copies of Ian Fleming books as a tween, and as an adult, I enjoy the crazy poetic license of the movies and join in the debate about my favorite Bond actor. I particularly enjoy the Q character, who appeals to my nerdiness. Unfortunately, Q's exploits were static, and did not lend themselves to mapping. The map of 007's exploit locations, and two additional metrics, shown in the e-Poster, and based on the SGMAP sample, will have to do.

CONCLUSION

SAS has always been at the forefront of data visualization, from the days of SAS/GRAPH to the SG procedures, SAS Visual Analytics, and SAS Viya. The use of annotation and attributes can greatly enhance both SAS/GRAPH and ODS GRAPHICS output and allow a high degree of customization. I hope you will look at the newish PROC SGMAP procedure for following your data (or guilty pleasures) and have an increased understanding of how these similar, but different, procedures work.

REFERENCES

- Allison, R. (2019). "What's in the pipeline for the SGMAP Procedure?!?". Proceedings of South East SAS Users Group Conference, Williamsburg, Virginia. Retrieved from https://www.lexjansen.com/sesug/2019/SESUG2019_Paper-292_Final_PDF.pdf
- Allison, R., Hadden, L. and Zdeb, M. (2008). "Wow! You Did That Map with SAS/GRAPH®?". Proceedings of Northeast SAS Users Group Conference, Pittsburgh, Pennsylvania. Retrieved from <https://www.lexjansen.com/nesug/nesug08/po/po01.pdf>
- Andrews, R. and Allison, R. (2015). "Creating Geographic Rating Area Maps: How to Combine Counties, Split Counties, and use Zip Code Boundaries". Proceedings of South East SAS Users Group Conference, Savannah, Georgia. Retrieved from <https://www.lexjansen.com/sesug/2015/RV-204.pdf>
- Andrews, R., Hadden, L. and Allison, R. (2017). "Methods for Creating Sparklines using SAS®". Proceedings of South East SAS Users Group Conference, Cary, North Carolina. Retrieved from https://analytics.ncsu.edu/sesug/2017/SESUG2017_Paper-42_Final_PDF.pdf
- Bessler, L. (1997). "Map Smart: Design and Build Effective InfoGeographics Using PROC GMAP and Software Intelligence". Proceedings of Midwest SAS Users Group Conference, Chicago, Illinois. Retrieved from <https://www.lexjansen.com/mwsug/1997/MWSUG97022.pdf>
- Bessler, L. 1996 "Better Low-Cost InfoGeographics Today: Improved Readability and Usability for PROC GMAP Output". Proceedings of SAS Users Group International Conference, Chicago, Illinois. Retrieved from <https://www.lexjansen.com/sugi/sugi21/iv/143-21.pdf>
- Bessler, L. 2001 "At Home and Abroad with PROC GMAP: A Global Atlas of InfoGeographic Design For Effective Visual Communication". Proceedings of SAS Users Group International Conference, Long Beach, California. Retrieved from <https://support.sas.com/resources/papers/proceedings/proceedings/sugi26/p192-26.pdf>
- Cartier, J. and Heath, Dan. (2007). "Using ODS Styles with SAS/GRAPH®". Proceedings of SAS GLOBAL FORUM Conference, Orlando, Florida. Retrieved from <https://support.sas.com/resources/papers/proceedings/proceedings/forum2007/088-2007.pdf>
- Fernandes, A. (2011). "Case Study: Geospatial location analysis & reporting application using SAS, Google Maps & MS Excel". Proceedings of Midwest SAS Users Group Conference, Kansas City, Kansas. Retrieved from <https://www.lexjansen.com/mwsug/2011/dataviz/MWSUG-2011-DG01.pdf>
- Hadden, L. (2003). "What's in a Map? A Macro-Driven Drill-Down Geo-Graphical Representation system". Proceedings of Western Users of SAS Software Conference, San Francisco, California. Retrieved from https://www.lexjansen.com/wuss/2003/DataPresentation/i-whats_in_a_map.pdf

Hadden, L. (2009). "Behind the Scenes with SAS®: Using Custom Graphics in SAS® Output". Proceedings of SAS GLOBAL FORUM Conference, National Harbor, Maryland. Retrieved from <https://support.sas.com/resources/papers/proceedings09/185-2009.pdf>

Hadden, L. (2011). "Programming the Provider Previews: Extreme SAS® Reporting". Proceedings of Northeast SAS Users Group Conference, Portland, Maine. Retrieved from <https://www.lexjansen.com/nesug/nesug11/gr/gr06.pdf>

Hadden, L. (2013). "Extreme SAS reporting II: Data Compendium and 5 Star Ratings Revisited". Proceedings of Northeast SAS Users Group Conference, Burlington, Vermont. Retrieved from https://www.lexjansen.com/nesug/nesug13/122_Final_Paper.pdf

Hadden, L. (2013). "Where in the World are SAS/Graph Maps? An Exploration of the Old and New SAS Mapping Capacities". Proceedings of Northeast SAS Users Group Conference, Burlington, Vermont. Retrieved from https://www.lexjansen.com/nesug/nesug13/124_Final_Paper.pdf

Hadden, L. (2016). "Red Rover, Red Rover, Send Data Right Over: Exploring External Geographic Data Sources with SAS®". Proceedings of Midwest SAS Users Group Conference, Cincinnati, Ohio. Retrieved from <https://www.lexjansen.com/mwsug/2016/DV/MWSUG-2016-DV05.pdf>

Hadden, L. (2017). "SAS/GRAPH® and GfK GeoMarketing Maps: a Subject Matter Expert Winning Combination". Proceedings of SAS GLOBAL FORUM Conference, Orlando, Florida. Retrieved from <https://support.sas.com/resources/papers/proceedings17/1311-2017.pdf>

Hadden, L. (2018). "Wow! You Did That Map With SAS®?! Round II". Proceedings of Western Users of SAS Software Conference, Sacramento, California. Retrieved from https://www.lexjansen.com/wuss/2018/32_Final_Paper_PDF.pdf

Hadden, L. (2019). "Using ODS Trace (DOM), Procedural Output and ODS Output Objects to Create the Output of Your Dreams". Proceedings of Midwest SAS Users Group Conference, Chicago, Illinois. Retrieved from <https://www.lexjansen.com/mwsug/2019/SP/MWSUG-2019-SP-057.pdf>

Hadden, L. (2021). "Dressing Up your SAS/GRAPH and SG Procedural Output with Templates, Attributes and Annotation". Proceedings of South East SAS Users Group Conference, Virtual Conference. Retrieved from https://www.lexjansen.com/sesug/2021/SESUG2021_Paper_59_Final_PDF.pdf

Hadden, L. (2021). "Visually Exploring Proximity Analyses Using SAS PROC GEOCODE and SGMAP and Public Use Data Sets". Proceedings of Pharmaceutical SAS Users Group Conference, Virtual Conference. Retrieved from <https://www.lexjansen.com/pharmasug/2021/DV/PharmaSUG-2021-DV-188.pdf>

Hadden, L. (2023). "A Deep Dive into Enhancing SAS/GRAPH and SG Procedural Output with Templates, Styles, Attributes, and Annotation". Proceedings of Western Users of SAS Software Conference, San Diego, California. Retrieved from <https://www.lexjansen.com/wuss/2023/WUSS-2023-Paper-136.pdf>

Hadden, L. (2023). "SAS® PROC GEOCODE by Example: A Case Study". Proceedings of SAS EXPLORE Conference, Las Vegas, Nevada. Retrieved from <https://communities.sas.com/t5/SAS-Explore-Presentations/SAS-PROC-GEOCODE-by-Example-A-Case-Study/ta-p/896670>

Hadden, L. (2024). "A Deep Dive into Enhancing SAS/GRAPH® and SG Procedural Output with Templates, Styles, Attributes, and Annotation". Proceedings of Pharmaceutical SAS Users Group Conference, Baltimore, Maryland. Retrieved from <https://www.lexjansen.com/pharmasug/2024/PO/PharmaSUG-2024-PO-128.pdf>

Hadden, L. 1997 "From 50,000,000 Claims to One Analytical File". Proceedings of SAS Users Group International Conference, San Diego, California. Retrieved from <https://support.sas.com/resources/papers/proceedings/proceedings/sugi22/POSTERS/PAPER210.PDF>

Hadden, L. and Zdeb, M. (2010). "ZIP Code 411: Decoding SASHELP.ZIPCODE and Other SAS® Maps Online Mysteries". Proceedings of Northeast SAS Users Group Conference, Baltimore, Maryland. Retrieved from <https://www.lexjansen.com/nesug/nesug10/po/po10.pdf>

Hadden, L., Johnson, A. and Olsho, L. (2007). "Color Your World - With SAS®". Proceedings of Northeast SAS Users Group Conference, Baltimore, Maryland. Retrieved from <https://www.lexjansen.com/nesug/nesug07/po/po11.pdf>

Havens-McColgan, J. (2012). "Applications of PROC GEOCODE and Incorporation of Census Block-Level Data". Proceedings of SAS GLOBAL FORUM Conference, Orlando, Florida. Retrieved from <https://support.sas.com/resources/papers/proceedings12/118-2012.pdf>

Heath, D. (2007). "New SAS/GRAPH® Procedures for Creating Statistical Graphics in Data Analysis". Proceedings of SAS GLOBAL FORUM Conference, Orlando, Florida. Retrieved from <https://support.sas.com/resources/papers/proceedings/proceedings/forum2007/193-2007.pdf>

Heath, D. (2008). "Effective Graphics Made Simple Using SAS/GRAPH® SG Procedures". Proceedings of Pharmaceutical SAS Users Group Conference, Atlanta, Georgia. Retrieved from <https://www.lexjansen.com/pharmasug/2008/sas/SA06.pdf>

Heath, D. (2009). "Secrets of the SG Procedures". Proceedings of SAS GLOBAL FORUM Conference, National Harbor, Maryland. Retrieved from <https://support.sas.com/resources/papers/proceedings09/324-2009.pdf>

Heath, D. (2010). "Creating Presentation-Quality ODS Graphics Output". Proceedings of SAS GLOBAL FORUM Conference, Seattle, Washington. Retrieved from <https://support.sas.com/resources/papers/proceedings10/237-2010.pdf>

Heath, D. (2011). "Now You Can Annotate Your Statistical Graphics Procedure Graphs". Proceedings of Western Users of SAS Software Conference, San Francisco, California. Retrieved from https://www.lexjansen.com/wuss/2011/datapresentation/Papers_Heath_D_76206.pdf

Heath, D. (2014). "Putting on the Ritz: New Ways to Style Your ODS Graphics to the Max". Proceedings of Midwest SAS Users Group Conference, Chicago, Illinois. Retrieved from <https://www.lexjansen.com/mwsug/2014/SS/MWSUG-2014-SS02.pdf>

Heath, D. (2016). "Annotating the SAS® ODS Graphics Way!". Proceedings of SAS GLOBAL FORUM Conference, Las Vegas, Nevada. Retrieved from <https://support.sas.com/resources/papers/proceedings16/SAS5300-2016.pdf>

Heath, D. (2016). "Now You Can Annotate Your GTL Graphs!". Proceedings of Pharmaceutical SAS Users Group Conference, Denver, Colorado. Retrieved from <https://www.lexjansen.com/pharmasug/2016/DG/PharmaSUG-2016-DG01.pdf>

Heath, D. (2018). "Diving Deep into SAS® ODS Graphics Styles". Proceedings of Pharmaceutical SAS Users Group Conference, Seattle, Washington. Retrieved from <https://www.lexjansen.com/pharmasug/2018/DV/PharmaSUG-2018-DV02.pdf>

Heath, D. (2024). "Building Complex Graphics from Simple Plot Types". Proceedings of Pharmaceutical SAS Users Group Conference, Baltimore, Maryland. Retrieved from <https://www.lexjansen.com/pharmasug/2024/HT/PharmaSUG-2024-HT-197.pdf>

Hebbar, P., Li, L. and Matange, S. (2016). "Graph a Million with the SGPLOT Procedure". Proceedings of SAS GLOBAL FORUM Conference, Las Vegas, Nevada. Retrieved from <https://support.sas.com/resources/papers/proceedings16/SAS4341-2016.pdf>

Hughes, T. M. (2013). "Binning Bombs When You're Not a Bomb Maker: A Code-Free Methodology to Standardize, Categorize, and Denormalize Categorical Data". Proceedings of South East SAS Users Group Conference, St. Pete Beach, Florida. Retrieved from <https://analytics.ncsu.edu/sesug/2013/BB-04.pdf>

Hughes, T. M. (2018). "Winning the War on Terror with Waffles: Maximizing GINSIDE Efficiency for Blue Force Tracking Big Data". Proceedings of South Central SAS Users Group Conference, Austin, Texas. Retrieved from <https://www.lexjansen.com/scsug/2018/rs29.pdf>

Hughes, T. M. (2022). "GIS Challenges of Cataloging Catastrophes: Serving up GeoWaffles with a Side of Hash Tables to Conquer Big Data Point-in-Polygon Determination and Supplant SAS® PROC

GINSIDE". Proceedings of Western Users of SAS Software Conference, San Francisco, California. Retrieved from <https://www.lexjansen.com/wuss/2022/WUSS-2022-Paper-79.pdf>

Ivis, F. (2006). "Calculating Geographic Distance: Concepts and Methods". Proceedings of Northeast SAS Users Group Conference, Philadelphia, Pennsylvania. Retrieved from <https://www.lexjansen.com/nesug/nesug06/dm/da15.pdf>

Kreuter, W. (1998). "Distance between Two Geographic Locations". Proceedings of Pacific Northwest SAS Users Group Conference, Portland, Oregon. Retrieved from <https://www.lexjansen.com/pnwsug/1998/PNWSUG98022.pdf>

LaSelva, G. D. (2010). "Aggregation without Aggravation: Determining Spatial Contiguity and Joining Geographic Areas Using Hashing". Proceedings of SAS GLOBAL FORUM Conference, Seattle, Washington. Retrieved from <https://support.sas.com/resources/papers/proceedings10/223-2010.pdf>

Lafler, R. P. and Wade, A. (2023). "From Interactive Mapmaking to Beautiful Geospatial Visualizations: Harnessing the Power of Python and Google Earth Engine for Extracting, Analyzing, and Visualizing High Resolution Spatiotemporal Data". Proceedings of Western Users of SAS Software Conference, San Diego, California. Retrieved from <https://www.lexjansen.com/wuss/2023/WUSS-2023-Paper-154.pdf>

Li, L. (2020). "Visualizing Geographical Data with a Tile Grid Map in SAS". Proceedings of SAS GLOBAL FORUM Conference, Virtual Conference. Retrieved from <https://support.sas.com/resources/papers/proceedings20/4335-2020.pdf>

Lin, A. Z. (2016). "Extracting Email Domains and Geo-Processing IP Addresses in SAS®". Proceedings of South East SAS Users Group Conference, Bethesda, Maryland. Retrieved from https://analytics.ncsu.edu/sesug/2016/EPO-193_Final_PDF.pdf

Massengill, D. (2010). "Google Maps and SAS/GRAPH®". Proceedings of Midwest SAS Users Group Conference, Milwaukee, Wisconsin. Retrieved from <https://www.lexjansen.com/mwsug/2010/appdev/MWSUG-2010-29.pdf>

Massengill, D. (2010). "Tips and Tricks: More SAS/GRAPH® Map Secrets". Proceedings of Midwest SAS Users Group Conference, Milwaukee, Wisconsin. Retrieved from <https://www.lexjansen.com/mwsug/2010/dataviz/MWSUG-2010-134.pdf>

Massengill, D. (2012). "Together at Last: Spatial Analysis and SAS® Mapping". Proceedings of SAS GLOBAL FORUM Conference, Orlando, Florida. Retrieved from <https://support.sas.com/resources/papers/proceedings12/284-2012.pdf>

Massengill, D. (2013). "'Google-like' Maps in SAS". Proceedings of South East SAS Users Group Conference, St. Pete Beach, Florida. Retrieved from <https://analytics.ncsu.edu/sesug/2013/RIV-14.pdf>

Massengill, D. (2016). "The New SAS® Map Data Sets". Proceedings of SAS GLOBAL FORUM Conference, Las Vegas, Nevada. Retrieved from <https://support.sas.com/resources/papers/proceedings16/SAS3421-2016.pdf>

Massengill, D. and Odom, E. (2009). "PROC GEOCODE: Creating Map Locations from Your Data". Proceedings of SAS GLOBAL FORUM Conference, National Harbor, Maryland. Retrieved from <https://support.sas.com/resources/papers/proceedings09/087-2009.pdf>

Massengill, D. and Odom, E. (2010). "PROC GEOCODE: Now with Street-Level Geocoding". Proceedings of SAS GLOBAL FORUM Conference, Seattle, Washington. Retrieved from <https://support.sas.com/resources/papers/proceedings10/332-2010.pdf>

Massengill, D. and Odom, E. (2013). "PROC GEOCODE: Finding Locations Outside the U.S.". Proceedings of SAS GLOBAL FORUM Conference, San Francisco, California. Retrieved from <https://support.sas.com/rnd/papers/sasgf13/Geocode2013.pdf>

Matange, S. (2015). "Graphs Are Easy with SAS® 9.4". Proceedings of SAS GLOBAL FORUM Conference, Dallas, TX. Retrieved from <https://support.sas.com/resources/papers/proceedings15/SAS1780-2015.pdf>

- Mohan, P. (2020). "Practical Geospatial Analysis of Open and Public-Use Data". Proceedings of SAS GLOBAL FORUM Conference, Virtual Conference. Retrieved from <https://support.sas.com/resources/papers/proceedings20/4293-2020.pdf>
- Moser, E. B. (2002). "Geostatistical Tools in the SAS System". Proceedings of South Central SAS Users Group Conference, Richardson, Texas. Retrieved from <https://www.lexjansen.com/scsug/2002/SCSUG02034.pdf>
- Nguyen, M., and Trahan S. (2009). "Geospatial Analysis Using SAS® and the Google Map API". Proceedings of SAS Global Forum Conference, National Harbor, Maryland. Retrieved from <https://support.sas.com/resources/papers/proceedings09/015-2009.pdf>
- Okerson, B. (2003). "Focusing Healthcare Quality Improvement Dollars: Using SAS for Geographic Targeting". Proceedings of Northeast SAS Users Group Conference, Washington, District of Columbia. Retrieved from <https://www.lexjansen.com/nesug/nesug03/gr/gr002.pdf>
- Okerson, B. (2011). "Distance Mapping in Health and Healthcare: SAS® as a Tool for Health Geomatics". Proceedings of SAS GLOBAL FORUM Conference, Las Vegas, Nevada. Retrieved from <https://support.sas.com/resources/papers/proceedings11/299-2011.pdf>
- Okerson, B. (2014). "Time Series Mapping with SAS®: Visualizing Geographic Change over Time in the Health Insurance Industry". Proceedings of SAS GLOBAL FORUM Conference, Washington, DC. Retrieved from <https://support.sas.com/resources/papers/proceedings14/1422-2014.pdf>
- Osterloo, K. and Wu, A. (2017). "Geospatial Analysis: Linear, Nonlinear, or Both?". Proceedings of SAS GLOBAL FORUM Conference, Orlando, Florida. Retrieved from <https://support.sas.com/resources/papers/proceedings17/1128-2017.pdf>
- Sa, T. (2018). "A Macro that Can Get the Geo Coding Information from the Google Map API". Proceedings of Midwest SAS Users Group Conference, Indianapolis, Indiana. Retrieved from <https://www.lexjansen.com/mwsug/2018/SP/MWSUG-2018-SP-57.pdf>
- Sa, T. (2018). "Macro that Can Get Geo Coding Information from the Google Maps API". Proceedings of SAS GLOBAL FORUM Conference, Denver, Colorado. Retrieved from <https://support.sas.com/resources/papers/proceedings18/2518-2018.pdf>
- Schulz, F. (2011). "Interactive Geospatial Analysis with SAS Portal: How to Use SAS® Real-Time Processes to Dynamically Integrate with Google Maps, ESRI Maps, and Open Street Maps". Proceedings of SAS GLOBAL FORUM Conference, Las Vegas, Nevada. Retrieved from <https://support.sas.com/resources/papers/proceedings11/276-2011.pdf>
- Simon, L. and Massengill, D. (2013). "Tips and Tricks: Using the new SAS Map data sets". Proceedings of SAS GLOBAL FORUM Conference, San Francisco, California. Retrieved from <https://support.sas.com/rnd/papers/sasgf13/GfKMaps2013.pdf>
- Smith, L. (2018). "Macro Method to Use Google Maps and SAS® to Geocode a Location by Name or Address". Proceedings of SAS GLOBAL FORUM Conference, Denver, Colorado. Retrieved from <https://support.sas.com/resources/papers/proceedings18/2684-2018.pdf>
- Sorrentino, K. and Plano, J. (2023). "Geocoding in SAS®: The Basics". Proceedings of South East SAS Users Group Conference, Charlotte, North Carolina. Retrieved from https://www.lexjansen.com/sesug/2023/SESUG2023_Paper_110_Final_PDF.pdf
- Spicer, J. (1998). "Delivering Geographic Information: Just Click My Map Object". Proceedings of Northeast SAS Users Group Conference, Pittsburgh, Pennsylvania. Retrieved from <https://www.lexjansen.com/nesug/nesug98/solu/p096.pdf>
- Spicer, J. (1998). "Delivering Geographic Information: For Those Who Can't Read a GMAP and Won't Stop to Ask for Directions". Proceedings of SAS Users Group International Conference, Nashville, Tennessee. Retrieved from <https://support.sas.com/resources/papers/proceedings/proceedings/sugi23/Infovis/p143.pdf>

Summers, E., Allison, R., Langston, J. and Cowley, J. (2012). "Using SAS/GRAPH® to Create Visualizations That Also Support Tactile and Auditory Interaction". Proceedings of SAS GLOBAL FORUM Conference, Orlando, Florida. Retrieved from <https://support.sas.com/resources/papers/proceedings12/279-2012.pdf>

Walker, C. (2016). "Put Your Data on the Map: Using the GEOCODE and GMAP Procedures to Create Bubble Maps in SAS®". Proceedings of SAS GLOBAL FORUM Conference, Las Vegas, Nevada. Retrieved from <https://support.sas.com/resources/papers/proceedings16/10404-2016.pdf>

Watson, R. and Hadden, L. (2021). "'Bored'-Room Buster Bingo - Create Bingo Cards Using SAS® ODS Graphics". Proceedings of South East SAS Users Group Conference, Virtual Conference. Retrieved from https://www.lexjansen.com/sesug/2021/SESUG2021_Paper_38_Final_PDF.pdf

Williams, C. and Hadden, L. (2015). "PROC RANK, PROC SQL, PROC FORMAT and PROC GMAP Team Up and a (Map) Legend is Born!". Proceedings of South East SAS Users Group Conference, Savannah, Georgia. Retrieved from <https://analytics.ncsu.edu/sesug/2015/RV-80.pdf>

Zdeb, M. and Allison, R. (2005). "Stretching the Bounds of SAS/GRAPH® Software". Proceedings of Northeast SAS Users Group Conference, Portland, Maine. Retrieved from <https://www.lexjansen.com/nesug/nesug05/dp/dp5.pdf>

Zdeb, M. and Allison, R. (2006). "SAS/GRAPH® 101". Proceedings of SAS Users Group International Conference, San Francisco, California. Retrieved from <https://support.sas.com/resources/papers/proceedings/proceedings/sugi31/239-31.pdf>

ACKNOWLEDGMENTS

To Robert Allison and Mike Zdeb, frequent co-authors and brilliant minds, a big thank you for all the fantastic projects we engaged in.

To Troy Martin Hughes and his Geowaffles, another huge thank you for being my GeoMuse.

To LeRoy Bessler, fellow mapmaker and author, who inspired me with his graphic outputs that danced and sung more than a decade ago, a hearty thank you.

To Sanjay Matange and Dan Heath, who answered so many of my questions, thanks for all the fish!

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Louise S. Hadden
Cormac Corporation
Louisesquibbhadden@gmail.com

Any brand and product names are trademarks of their respective companies.

Any opinions expressed in this document are my own.